NAG Library Routine Document

F08KNF (ZGELSS)

Note: before using this routine, please read the Users' Note for your implementation to check the interpretation of *bold italicised* terms and other implementation-dependent details.

1 Purpose

F08KNF (ZGELSS) computes the minimum norm solution to a complex linear least squares problem

 $\min_{x} \|b - Ax\|_2.$

2 Specification

SUBROUTINE F08KNF (M, N, NRHS, A, LDA, B, LDB, S, RCOND, RANK, WORK, & LWORK, RWORK, INFO)

INTEGER M, N, NRHS, LDA, LDB, RANK, LWORK, INFO REAL (KIND=nag_wp) S(*), RCOND, RWORK(*) COMPLEX (KIND=nag_wp) A(LDA,*), B(LDB,*), WORK(max(1,LWORK))

The routine may be called by its LAPACK name zgelss.

3 Description

F08KNF (ZGELSS) uses the singular value decomposition (SVD) of A, where A is an m by n matrix which may be rank-deficient.

Several right-hand side vectors b and solution vectors x can be handled in a single call; they are stored as the columns of the m by r right-hand side matrix B and the n by r solution matrix X.

The effective rank of A is determined by treating as zero those singular values which are less than RCOND times the largest singular value.

4 References

Anderson E, Bai Z, Bischof C, Blackford S, Demmel J, Dongarra J J, Du Croz J J, Greenbaum A, Hammarling S, McKenney A and Sorensen D (1999) *LAPACK Users' Guide* (3rd Edition) SIAM, Philadelphia http://www.netlib.org/lapack/lug

Golub G H and Van Loan C F (1996) Matrix Computations (3rd Edition) Johns Hopkins University Press, Baltimore

5 **Parameters**

1: M – INTEGER

On entry: m, the number of rows of the matrix A. Constraint: M > 0.

2: N – INTEGER

On entry: n, the number of columns of the matrix A. Constraint: $N \ge 0$. Input

Input

Input

Input

Input/Output

Input/Output

3: NRHS – INTEGER

On entry: r, the number of right-hand sides, i.e., the number of columns of the matrices B and X. Constraint: NRHS ≥ 0 .

4: A(LDA, *) – COMPLEX (KIND=nag_wp) array

Note: the second dimension of the array A must be at least max(1, N).

On entry: the m by n matrix A.

On exit: the first $\min(m, n)$ rows of A are overwritten with its right singular vectors, stored row-wise.

5: LDA – INTEGER

On entry: the first dimension of the array A as declared in the (sub)program from which F08KNF (ZGELSS) is called.

Constraint: LDA $\geq \max(1, M)$.

6: B(LDB, *) – COMPLEX (KIND=nag_wp) array

Note: the second dimension of the array B must be at least max(1, NRHS).

On entry: the m by r right-hand side matrix B.

On exit: B is overwritten by the n by r solution matrix X. If $m \ge n$ and RANK = n, the residual sum of squares for the solution in the *i*th column is given by the sum of squares of the modulus of elements $n + 1, \ldots, m$ in that column.

7: LDB – INTEGER

On entry: the first dimension of the array B as declared in the (sub)program from which F08KNF (ZGELSS) is called.

Constraint: LDB $\geq \max(1, M, N)$.

8: S(*) - REAL (KIND=nag_wp) array

Note: the dimension of the array S must be at least max(1, min(M, N)).

On exit: the singular values of A in decreasing order.

9: RCOND – REAL (KIND=nag_wp)

On entry: used to determine the effective rank of A. Singular values $S(i) \le RCOND \times S(1)$ are treated as zero. If RCOND < 0, *machine precision* is used instead.

10: RANK – INTEGER

On exit: the effective rank of A, i.e., the number of singular values which are greater than $RCOND \times S(1)$.

- WORK(max(1,LWORK)) COMPLEX (KIND=nag_wp) array Workspace
 On exit: if INFO = 0, the real part of WORK(1) contains the minimum value of LWORK required for optimal performance.
- 12: LWORK INTEGER

On entry: the dimension of the array WORK as declared in the (sub)program from which F08KNF (ZGELSS) is called.

Output

Input

Input

Input

Output

If LWORK = -1, a workspace query is assumed; the routine only calculates the optimal size of the WORK array, returns this value as the first entry of the WORK array, and no error message related to LWORK is issued.

Suggested value: for optimal performance, LWORK should generally be larger. Consider increasing LWORK by at least $nb \times \min(M, N)$, where nb is the optimal **block size**.

 $\textit{Constraint: LWORK} \geq 1 ~ and ~ LWORK \geq 2 \times min(M,N) + max(M,N,NRHS).$

13: RWORK(*) – REAL (KIND=nag_wp) array

Note: the dimension of the array RWORK must be at least $max(1, 5 \times min(M, N))$.

14: INFO – INTEGER

Output

Workspace

On exit: INFO = 0 unless the routine detects an error (see Section 6).

6 Error Indicators and Warnings

INFO < 0

If INFO = -i, argument *i* had an illegal value. An explanatory message is output, and execution of the program is terminated.

INFO > 0

The algorithm for computing the SVD failed to converge; if INFO = i, *i* off-diagonal elements of an intermediate bidiagonal form did not converge to zero.

7 Accuracy

See Section 4.5 of Anderson et al. (1999) for details.

8 Parallelism and Performance

F08KNF (ZGELSS) is threaded by NAG for parallel execution in multithreaded implementations of the NAG Library.

F08KNF (ZGELSS) makes calls to BLAS and/or LAPACK routines, which may be threaded within the vendor library used by this implementation. Consult the documentation for the vendor library for further information.

Please consult the X06 Chapter Introduction for information on how to control and interrogate the OpenMP environment used within this routine. Please also consult the Users' Note for your implementation for any additional implementation-specific information.

9 Further Comments

The real analogue of this routine is F08KAF (DGELSS).

10 Example

This example solves the linear least squares problem

 $\min \|b - Ax\|_2$

for the solution, x, of minimum norm, where

$$A = \begin{pmatrix} 0.47 - 0.34i & -0.40 + 0.54i & 0.60 + 0.01i & 0.80 - 1.02i \\ -0.32 - 0.23i & -0.05 + 0.20i & -0.26 - 0.44i & -0.43 + 0.17i \\ 0.35 - 0.60i & -0.52 - 0.34i & 0.87 - 0.11i & -0.34 - 0.09i \\ 0.89 + 0.71i & -0.45 - 0.45i & -0.02 - 0.57i & 1.14 - 0.78i \\ -0.19 + 0.06i & 0.11 - 0.85i & 1.44 + 0.80i & 0.07 + 1.14i \end{pmatrix}$$

and

$$b = \begin{pmatrix} -1.08 - 2.59i \\ -2.61 - 1.49i \\ 3.13 - 3.61i \\ 7.33 - 8.01i \\ 9.12 + 7.63i \end{pmatrix}.$$

A tolerance of 0.01 is used to determine the effective rank of A.

Note that the block size (NB) of 64 assumed in this example is not realistic for such a small problem, but should be suitable for large problems.

10.1 Program Text

Program f08knfe

```
!
     FO8KNF Example Program Text
     Mark 25 Release. NAG Copyright 2014.
1
1
      .. Use Statements ..
     Use nag_library, Only: dznrm2, nag_wp, zgelss
1
      .. Implicit None Statement ..
     Implicit None
      .. Parameters ..
!
     Integer, Parameter
                                      :: nb = 64, nin = 5, nout = 6
     .. Local Scalars ..
!
     Real (Kind=nag_wp)
                                       :: rcond, rnorm
     Integer
                                       :: i, info, lda, lwork, m, n, rank
!
      .. Local Arrays ..
     Complex (Kind=nag_wp), Allocatable :: a(:,:), b(:), work(:)
     Real (Kind=nag_wp), Allocatable :: rwork(:), s(:)
      .. Executable Statements ..
1
     Write (nout,*) 'FO8KNF Example Program Results'
     Write (nout,*)
     Skip heading in data file
!
     Read (nin,*)
     Read (nin,*) m, n
     lda = m
     lwork = 2*n + nb*(m+n)
     Allocate (a(lda,n),b(m),work(lwork),rwork(5*n),s(n))
!
     Read A and B from data file
     Read (nin,*)(a(i,1:n),i=1,m)
     Read (nin,*) b(1:m)
     Choose RCOND to reflect the relative accuracy of the input data
1
     rcond = 0.01E0_nag_wp
1
     Solve the least squares problem min( norm2(b - Ax) ) for the x
     of minimum norm.
1
1
     The NAG name equivalent of zgelss is f08knf
     Call zgelss(m,n,1,a,lda,b,m,s,rcond,rank,work,lwork,rwork,info)
      If (info==0) Then
!
       Print solution
```

```
Write (nout,*) 'Least squares solution'
        Write (nout,99999) b(1:n)
1
        Print the effective rank of A
        Write (nout,*)
        Write (nout,*) 'Tolerance used to estimate the rank of A'
        Write (nout,99998) rcond
        Write (nout,*) 'Estimated rank of A'
        Write (nout,99997) rank
!
        Print singular values of A
        Write (nout,*)
        Write (nout,*) 'Singular values of A'
        Write (nout,99996) s(1:n)
1
        Compute and print estimate of the square root of the
1
        residual sum of squares
        If (rank==n) Then
!
          The NAG name equivalent of dznrm2 is f06jjf
          rnorm = dznrm2(m-n,b(n+1),1)
          Write (nout,*)
          Write (nout,*) 'Square root of the residual sum of squares'
          Write (nout,99998) rnorm
        End If
      Else
        Write (nout,*) 'The SVD algorithm failed to converge'
      End If
99999 Format (4(' (',F7.4,',',F7.4,')':))
99998 Format (3X,1P,E11.2)
99997 Format (1X,I6)
99996 Format (1X,7F11.4)
    End Program f08knfe
```

10.2 Program Data

FO8KNF Example Program Data

5 4 :Values of M and N (0.47,-0.34) (-0.40, 0.54) (0.60, 0.01) (0.80,-1.02) (-0.32,-0.23) (-0.05, 0.20) (-0.26,-0.44) (-0.43, 0.17) (0.35,-0.60) (-0.52,-0.34) (0.87,-0.11) (-0.34,-0.09) (0.89, 0.71) (-0.45,-0.45) (-0.02,-0.57) (1.14,-0.78) (-0.19, 0.06) (0.11,-0.85) (1.44, 0.80) (0.07, 1.14) :End of matrix A (-1.08,-2.59) (-2.61,-1.49) (3.13,-3.61) (7.33,-8.01) (9.12, 7.63) :End of vector b

10.3 Program Results

Estimated rank of A 3 Singular values of A 2.9979 1.9983 1.0044 0.0064